

Comparison of Two Means for Independent and Dependent Samples Using SAS Proc TTEST

Example - Independent Samples

An entomologist is interested in the effectiveness of two different types of moth traps. He places eleven (11) type 1 traps and eight (8) type 2 traps out during the night. Later, he counts the number of moths caught in each trap. The entomologist would like to know if there is a difference between the catch rate of the two traps. The data collected from this study are presented below.

TRAP 1 41 34 33 36 40 25 31 37 34 30 38

TRAP 2 52 57 62 55 64 57 56 55

SAS Code using Proc ttest

```
options pageno = 1;
title "Comparison of Two Types of Moth Traps";
data moths;
input Trap_Type Moths @@;
cards;
1 41 1 34 1 33 1 36 1 40 1 25 1 31 1 37 1 34 1 30 1 38
2 52 2 57 2 62 2 55 2 64 2 57 2 56 2 55
;
proc print data = moths;
run;
title2 "Analysis Using an Independent Two-Sample T-test";
proc ttest data = moths;
  class Trap_Type;
  var Moths;
run;
```

The components, with explanation, for the program, including Proc ttest for the comparison of two means from independent samples are:

```
options pageno = 1;
```

The options statement can go anywhere in SAS and is used to control some aspects of the output. In this instance the “pageno = 1” option forces SAS to number its pages starting with 1. If this were left off, SAS would have begun numbering from where it left off. Other options for this statement are *linesize = n* and *pagesize = n*, which are used to set the number of characters across a page and the number of lines on a page.

```
title "Comparison of Two Types of Moth Traps";
title2 "Analysis Using an Independent Two-Sample T-test";
```

Are the title and subtitle printed on each page of the output following their definition (the subtitle will not be printed for the above program until after the Proc Print has been run).

```
proc ttest data = moths;
```

(Invokes Proc ttest and tells SAS to use the “moths” data set.)

```
  class Trap_Type;
```

(Tells Proc ttest which variable to use to determine the two treatments or populations)

```
  var Moths;
run;
```

(Tells Proc ttest which variable is the response)